

Example model statements for PROC MIXED (or GLIMMIX). *No warranties.* Use as a guide only.  
There are various assumptions (generally not described) behind each model.  
No guarantee that statements are always typed correctly.

## Examples of some MIXED or GLIMMIX statements used for various experimental designs

I followed SAS convention, and put comments between */\** and *\*/*. This way, the user can (generally) cut-and-paste. If not specified, the code is for MIXED. Any code for GLIMMIX is explicitly given (usually). This is a work in progress. I add to, or modify, this file as the need arises, or when I get inspired.

```
/*
alpha, beta, gamma:      fixed-effects factors (class variables)
block, rep, sub, grp:    random-effects factors (class variables)
y:                        response (dependent) variable - continuous
e:                        implied residual error
int:                      intercept term (a variable = constant*1 always)
```

For all models it is assumed, unless indicated otherwise, that a class statement is used, with all the specified variables/factors listed.

Example: *\*/*

```
class alpha beta block sub grp;
```

```
/* ONE FIXED FACTOR, completely randomized */
model y = alpha;
```

```
/* TWO FIXED factors, crossed, completely randomized */

model y = alpha|beta;
```

```
/*or */
```

```
model y = alpha beta alpha*beta;
```

```
/* ONE RANDOM factor (plus residual) */
```

```
model y = ;
random grp;
covtest 0; /*--test whether the variance for grp is 0;
```

```
/* ONE FIXED factor, in random blocks (RCB) */
```

```
model y = alpha;
random block;
```

```
/* a redundant way to fit this model would be: */
```

```
model y = alpha;
random block;
repeated / subject=block*alpha;
```

*/\* the repeated statement specifies the residual error, which is the experimental error here.*

Example model statements for PROC MIXED (or GLIMMIX). *No warranties.* Use as a guide only.  
There are various assumptions (generally not described) behind each model.  
No guarantee that statements are always typed correctly.

The statement is just giving the default (and thus not needed). \*/

```
/* another equivalent way to do a RCB (when block variance is positive):  
*/
```

```
    model y = alpha;  
    repeated / sub=block type=CS;
```

```
/* for GLIMMIX, the above would be: */  
    model y = alpha;  
    random _residual_ / sub=block type=CS;
```

/\* Note for above, the subject is block (the cluster); since there are multiple treatments within block, one could define different structure for the variances of each treatment and the covariances (correlations) of pairs of treatments within blocks. Compound Symmetry (CS) specifies a single covariance for all pairs of treatments (= block variance), and a single residual variance. \*/

```
/* Random blocks and random group factor; no fixed-effect factors */  
    model y = ;           *<--no terms in model statement;  
    random block grp;      *<--for either MIXED or GLIMMIX;  
    covtest . 0;          *<--test whether group variance is >0;  
                          *^--above statement for GLIMMIX procedure only;
```

/\* **ONE FIXED factor, random blocks, and SUB-SAMPLING** within experimental units. Note: experimental unit is identified by combination of block and alpha \*/

```
    model y = alpha;  
    random block block*alpha;
```

/\* When there is no sub-sampling, block\*alpha IS the residual (the variation among the experimental units [plots]). You do not need to specify the residual with MIXED or GLIMMIX. But with sub-samples (i.e., multiple observations for each block-treatment combination [experimental plot]), you need to be explicit and specify the block\*alpha random effect to properly test for group effects. Otherwise, you would get a residual that would be a mixture of block\*alpha and the sampling variances, leading to incorrect tests. \*/

/\*alternatives: \*/

```
    model y = alpha;  
    random int / sub=block;  
    random int / subject=block*alpha;
```

```
    model y = alpha;  
    random int alpha / sub=block;
```

/\*the first of these makes it clear that the subjects are block and also block\*alpha (experimental unit). The second of these

Example model statements for PROC MIXED (or GLIMMIX). *No warranties.* Use as a guide only.  
There are various assumptions (generally not described) behind each model.  
No guarantee that statements are always typed correctly.

```
is a short-hand approach, with subjects given by 1*block (i.e.,
block), and alpha*block.
The "int" is for intercept (a column of "ones"); thus, there are
variance terms for block, block*alpha (=exper. error), and
e (=sub-sampling error here), the latter being implied
by the earlier model */

/* alternative: */

    model y = alpha;
    random block block*alpha;
    repeated sub / subject=block*alpha;

/* the repeated statement specifies the residual error. That is,
there are subsamples nested within experimental units, comprising
combinations of block and alpha. This is the default here. The
repeated statement is not needed (exists by default).
So-called sampling error. */

/* alternative, using the correlation structure: */
    model y = alpha;
    random block;
    repeated / subject=block*alpha type=cs;

/* one can use the repeated statement for the experimental
error term (instead of just the sampling error), if one
specifies compound symmetry (cs) type for the block*alpha subject.
As with other cases, the sampling error need not be written
out. CS gives both block*alpha and sampling error. In GLIMMIX,
one would use:*/
    random _residual_ / subject=block*alpha type=cs;

/* ONE FIXED factor, random blocks, no sub-samples, but with unequal
variances:*/

    model y = alpha;
    random block;
    repeated / group=alpha;

/* or with GLIMMIX */
    random _residual_ / group=alpha;

/* ONE FIXED factor, random blocks, sub-samples, and
heterogeneity of the residuals (separate SAMPLING variance for each alpha
level, but a single experimental error variance. */

    model y = alpha;
    random block block*alpha;
    repeated sub / subject=block*alpha group=alpha;

/* or */
    model y = alpha;
    random block block*alpha;
```

Example model statements for PROC MIXED (or GLIMMIX). *No warranties.* Use as a guide only.  
 There are various assumptions (generally not described) behind each model.  
 No guarantee that statements are always typed correctly.

```

repeated / subject=block*alpha*sub group=alpha;

/* or, with GLIMMIX */
model y = alpha;
random block block*alpha;
random _residual_ / subject=block*alpha*sub group=alpha;

/* ONE FIXED factor, random blocks, and TRUE replications (rep)
of each treatment in each block. No interaction of block & alpha */

model y = alpha / ddfm =KR;
random block ;
/* Above looks just like a RCBD and no true reps (or one true rep per
block.*/

/* alternative, which is OVER-parameterized: */

model y = alpha / ddfm =KR;
random block;
random rep / subject=block*alpha;

/*alternative, which is also OVER-parameterized: */

model y = alpha / ddfm = KR;
random block rep(block alpha);
/* in above two formulations, the residual is separated out into two
terms: block*alpha*rep and the rest (implied). The F test for alpha turns
out to be the same. Same SEs also. But, it is better not to use the over-
parameterized version. */

/* RCB, true reps, and INTERACTION of block and alpha: */

model y = alpha / ddfm =KR ;
random block block*alpha;

/* use likelihood ratio (difference of -2log(L) for this and
previous model to see if there is an interaction. Or use: */
model y = alpha / ddfm =KR;
random block block*alpha;
covtest . 0; *---test whether the block*alpha variance is 0;
*^--above statement is for GLIMMIX only.;

/* SPLIT PLOT (with alpha as whole, and beta as sub-plot),
arranged in blocks */

model y = alpha|beta / ddfm =KR;
random block block*alpha;

/* alternative: */

```

Example model statements for PROC MIXED (or GLIMMIX). *No warranties. Use as a guide only.*  
 There are various assumptions (generally not described) behind each model.  
 No guarantee that statements are always typed correctly.

```

model y = alpha|beta / ddfm =KR;
random block;
random int / subject=block*alpha;

/* or */

model y = alpha|beta / ddfm = Satterth;
random block;
repeated / subject=block*alpha type=cs;

/* the latter uses compound symmetry (cs) type to specify
the experimental error with repeated statement. Statement gives
both block*alpha variance (variation among large experimental units, also
= the covariance among observations within large experimental plots) and
residual (variation among small experimental units [subplots]). */

/* a redundant alternative method would be: */

model y = alpha|beta / ddfm = Satterth;
random block block*alpha;
repeated beta/ subject=block*alpha;

/* where repeated is used for specifying the experimental
error beta terms, which is the default residual (variation in Y in small
experimental units). */

/* SPLIT PLOT (alpha as whole, beta as sub-plot, arranged
in blocks, and SUBSAMPLES (sub) */

model y = alpha|beta / ddfm =KR;
random block block*alpha beta(block alpha);

/* alternative: */
model y = alpha|beta / ddfm =KR;
random block;
random int beta / subject=block*alpha;

/* the latter statement specifies experimental error for
the large experimental units (whole-plots: block*alpha ), and for the
small experimental units [sub-plots: (beta(block alpha) or
beta*alpha*block]. Residual is sampling error. */

/* SPLIT PLOT - alpha as whole, beta as sub-plot, arranged
in blocks, and TRUE replications of whole plots in each
block (rep). Assuming no interaction of block * alpha */

model y = alpha|beta / ddfm =KR ;
random block rep(block alpha);

```

Example model statements for PROC MIXED (or GLIMMIX). *No warranties.* Use as a guide only.  
There are various assumptions (generally not described) behind each model.  
No guarantee that statements are always typed correctly.

```
/* alternative: */

model y = alpha|beta / ddfm =KR;
random block;
random rep / subject= block*alpha;

/* If there is a block*alpha interaction, use: */

model y = alpha|beta / ddfm = KR;
random block block*alpha rep(block alpha);

/* SPLIT-SPLIT PLOT with blocks - alpha as whole, beta as
sub-plot, and gamma as sub-sub plot, arranged in blocks.*/

model y = alpha|beta|gamma / ddfm = KR ;
random block block*alpha beta(block alpha);

/* alternative: */

model y = alpha|beta|gamma / ddfm = KR ;
random block;
random int beta / sub=block*alpha;

/* REPEATED MEASURES, with alpha as treatment, beta as time,
and block. For so-called compound symmetry (like a
split plot): */

model y = alpha|beta / ddfm = KR ;
random block;
repeated / subject=block*alpha type=cs;

/* Try different TYPE options for different covariance
structures. Examples: ar(1), un, cs, arh(1), ante(1),
toep, toeph, hf, fa(1), fa1(1), etc. Compare with AIC
values. Choose the covariance matrix that gives the lowest
AIC. With this R, then look at fixed effects in depth.*/

/* REPEATED MEASURES split plot, with alpha as whole plot, beta as sub-
plot, and gamma as time, and block. For so-called compound symmetry (like
a split-split plot): */

model y = alpha|beta|gamma / ddfm = KR ;
random block block*alpha;
repeated / subject=block*alpha*beta type=cs;

/* Try different TYPE options for different covariance
structures. Examples: ar(1), un, cs, arh(1), ante(1),
toep, toeph, hf, fa(1), fa1(1), etc. Compare with AIC
values. In general, choose the covariance matrix that
gives the lowest AIC.*/
```

Example model statements for PROC MIXED (or GLIMMIX). *No warranties.* Use as a guide only.  
There are various assumptions (generally not described) behind each model.  
No guarantee that statements are always typed correctly.

```
/* COVARIANCE ANALYSIS. As an example, consider alpha as  
a class (treatment) variable, and beta as time. Assuming  
that there is a first-order autoregressive structure to  
the residuals, one can use: */
```

```
class alpha block beta;  
model y = alpha|time / ddfm=KR SOLUTION;  
random block;  
repeated beta / subject=block*alpha type=ar(1);
```

```
/* Note that time is not listed in the class statement, but  
is used in the model statement. The solution option is  
needed to get the actual intercept and slope estimates.  
Note: the last value for alpha and alpha*t is 0, because  
this is the full model. Use CONTRASTS to get intercepts  
and slopes for each level of alpha. If this was a 2-way  
crossed factorial (not over time), then just take out  
repeated statement. See chapter 4 in The Study of Plant Disease Epidemics  
(2007) by Madden, Hughes, and van den Bosch for much more on this. */
```